Automated Summarization of Risk Assessment of Discrete and Continuous Exposures in Epidemiological Studies using SAS ODS

Stuart Long, Westat, Durham, NC
Rebecca Darden, Westat, Durham, NC

Abstract

In epidemiological research, we frequently deal with case-control studies where the researcher wishes to investigate associations between the outcome and numerous potential exposures added to a predetermined set of base model covariates. The exposures may be continuous, dichotomous, or multi-level. Our macro runs a PROC LOGISTIC model with adjustments for other predetermined base model covariates. PROC LOGISTIC statistics related to the added exposure variable are summarized, including the creation of class variables for multi-level exposures if needed. The CMH option in PROC FREQ computes risk estimates for only 2x2 tables, and not for the more general *rx2* tables. Our solution for rx2 level exposure variables is to create datasets for each comparison of an exposure level against the referent exposure group. PROC FREQ is run on these individual datasets and then re-run for the entire *rx2* table to obtain the estimated overall linear association. The PROC LOGISTIC and PROC FREQ statistics are merged and stored in a dataset for ease of reporting the final summary set of statistics to the researcher. The macro is written to allow use within any operating system. Although the macro code is advanced, using the macro is simple for those with additional statistical reporting needs.

Introduction

A large case-control study can present the epidemiologist with a wealth of potential exposures to investigate. The exposures may be demographic, environmental, or medical factors, and they may be dichotomous, continuous, or categorical variables. There will be a set of baseline model covariates which are thought to be confounders for the outcome-exposure relationships being investigated. The researcher will wish to look at both crude and adjusted associations between the outcome and numerous potential exposures. For example, a researcher may wish to look at associations between farming exposures and respiratory diseases. The potential exposures could range from farming crops and animals, to pesticides used, to work practices; this could amount to literally hundreds of potential associations. For each exposure, we perform both adjusted analyses (such as logistic regressions) and crude analyses (such as frequencies and Cochran-Mantel-Haenszel (CMH) statistics), thus multiplying the number of analyses done and listings generated. In our macro, for each potential exposure, we perform the necessary adjusted logistic regressions as well as the crude analyses, and summarize the results for the client to allow them to focus on particular aspects of the study without wading through stacks of printouts.

For this summarization of results, we need to extract information from both PROC LOGISTIC and PROC FREQ results. Logistic regressions are run using PROC LOGISTIC for each potential exposure, adjusting for the baseline model covariates, and pertinent statistics (such as the estimated odds ratio and 95% confidence limits) are extracted. The accompanying crude statistics for each exposure are obtained using PROC FREQ. We need to compute and extract the cell counts and column percentages, and both an estimate of the overall linear association between the exposure and the outcome, and estimates of the risk at each level of the exposure. For dichotomous and continuous exposures, there is only one estimated odds ratio necessary. However, if the researcher wishes to model an exposure with more than two levels (i.e., we have an exposure by outcome table that is *rx2*, and we will have *r-1* comparisons to make against the referent group), then the crude analyses need to reflect this. Since the CMH option in PROC FREQ provides the risk estimates only in the case of a 2x2 table, and not for the more general case of *rx2* tables, we necessarily must create separate data sets for the *r-1* comparisons and run a PROC FREQ on each data set. In our macro, we automate the creation of the necessary datasets, the set-up and execution of the adjusted logistic regressions, the computation of the accompanying crude analyses, the extraction of the pertinent counts and statistics, and the summarization of adjusted and crude analysis results for the numerous potential outcome-exposure associations.

Coding Methods

This macro code offers the user a simplistic method for summarizing the crude and adjusted statistics for the researcher to review and report. The user is required to supply a minimum number of parameters required in the modeling, and then is returned a data set which contains all the desired statistics which can then be reported in whatever method the user wishes. The parameters supplied by the user include the original working data set, the dichotomous outcome variable, the exposure variable, the referent level of the exposure variable, and the list of additional base model covariates included in the model. All variables must be numeric variables with labels and formats. In the following description, macro variables and data sets within parentheses are for reference to the macro code in Appendix 1.

The macro begins by separating the list of additional base model exposures into separate macro variables (**__bv&__k**). An analysis data set is created by keeping only the observations where the outcome, as well as all the exposure variables, are recorded (**__logis**). The purpose for this is to ensure that the crude statistics are generated for the same number of observations as the adjusted statistics. Macro variables are then generated which contain the number of levels in the outcome variable (**__oc_kt**) and the exposure variable (**__exp_kt**), which was passed in the macro call (**__expo**). The values of each level in these two variables (**__oc&i, __exp&i**) as well as the formats (**__fmt&i**) are also placed in macro variables. It is important to set a macro variable to the value of the format of the referent level of the exposure variable (**__reffmt**) since the CLASS statement in the PROC LOGISTIC will require the formatted value to identify the referent level in a multi-level exposure variable. Next, the CMH crude association statistic between the outcome and exposure variable is determined (**__crudep**). This is followed by the assignment of column frequencies (**__cell3, __cell4**) and percents between the exposure and outcome variable. These column frequencies and percents are from 2x2 associations with either a dichotomous exposure variable or, in the case of an r-level exposure variable (r>2), the referent level crossed with each of the comparison levels. For each 2x2 comparison, a one-observation data set (**__out&i**) is created which contains all of the afore mentioned information. These data sets are then concatenated to create a data set which contains all the column frequencies and percents for the 2x2 or rx2 comparison of the exposure variable with the outcome variable (**__fstats**). If the exposure variable contains more than 2 levels, then the adjusted statistics are generated for a trend test of the exposure variable. These statistics are stored in a one-observation data set (**__trend**) for later concatenation with the data set containing the individual class statistics (**__stats**). An additional PROC LOGISTIC is executed to generate the adjusted statistics for the dichotomous exposure variable, or CLASS variables in the event that the exposure variable contains more than 2 levels. The adjusted statistics are retained in data sets (**__param, __odds**) as one observation for each comparison level of the exposure

variable (r-1 observations). These data sets are then merged together with the data set that contains the crude statistics (**__fstats**) from the r-1 2x2 comparisons to form a data set that contains an observation for each comparison level of the exposure variable as well as an observation for the referent level (**__cstats**). If the exposure variable contains r>2 levels, then the data set containing the adjusted statistics from the trend test of the exposure variable (**__trend**) are added as an additional observation into the data set containing the crude and adjusted CLASS statistics (**__cstats**). Finally, the macro concatenates the statistics generated and stored in the final data set (**__cstats**) with a summary data set that contains statistics from each previous call of the macro for other exposure variables (**__sumary**).

Example

In our example presented in Appendix 2, we show how the user would include the macro code in a program for use. After including the code for the macro, the user may call the macro (**%summary**). There are 5 parameters in the macro call: the working data set (**sasdset**) , the outcome variable (**WHEEZE**), the exposure variable (**STATE** or **WEIGHT**), the value of the referent level for the exposure variable (**0** or **2**), and the list of base model covariates (**&log_vars**). In our example, we present 10 base model covariates. To improve readability of the macro call, we suggest long lists of covariates are passed as the value of a macro variable (**log_vars**). The macro call returns a data set of crude and adjusted statistics for all of the macro calls performed (**summary**). The user may decide to use PROC PRINT or PROC REPORT to generate a summary table for the researcher to review, or may use a DATA NULL similar to the one we have included in Appendix 2. The data in the data set may also be exported from SAS into a spreadsheet such as Excel or any other external software, if the user desires.

Summary

The accompanying macro code, using ODS, alleviates the tedious programming and manual extraction associated with the creation of individual datasets for PROC FREQ, the running of necessary crude and adjusted analyses using PROC FREQ and PROC LOGISTIC, and the extraction of pertinent cell counts and statistics when we have a large case-control study with numerous exposure variables. It provides the client with a data set of pertinent summary information without overwhelming him or her with pages and pages of output. The user can then determine what method of outputting the data will be used for reporting the summary statistics to the researcher.

Appendix 1. Macro Code

```
%MACRO summary(__dset, __outcom, __expo,
               __ref, __basevs);
/* this macro must be placed at this point    */
/* in the code to avoid mismatch of the %END  */
/* statement.                                 */
%MACRO datasets;
  %DO i = 1 %TO &__exp_kt;
      __out&i
  %END;
%MEND datasets;


/* create macro variables for each of the     */
/* base variables specified                   */
/* as well as a counter for how many base     */
/* variables are passed into the macro.       */
  %LET __varkt=0;
  %LET __k=1;
  %LET __w=%SCAN(&__basevs,&__k);
  %DO %WHILE(&__w NE);
      %LET __bv&__k=&__w;
      %LET __varkt=&__k;
      %LET __k=%EVAL(&__k+1);
      %LET __w=%SCAN(&__basevs,&__k);
  %END;

  /* check to see if the user requested the   */
  /* exposure variable to be continuous.  If  */
  /* it is continuous, then set the __ctn     */
  /* macro flag to '1' and reset the exposure */
  /* variable to be equal to __expo minus the */
  /* included pound sign (#).                 */
   %LET __pound=%SUBSTR(&__expo,1,1);
  %IF &__pound=# %THEN %DO;
      %LET __ctn=1;
      %LET __expo=%SUBSTR(&__expo,2);
  %END;
  %ELSE %LET __ctn=0;

/* create an analysis data set that contains  */
/* only the observations that will be in the  */
/* model.  Also use this data step to assign  */
/* the label and formats for the exposure     */
/* variable to macro variables.               */
DATA __logis;
    SET &__dset
     (KEEP = &__outcom &__expo &__basevs);
    IF _N_=1 THEN DO;
       CALL SYMPUT('__explab',VLABEL(&__expo));
       CALL SYMPUT('__oclab',VLABEL(&__outcom));
    END;
    IF &__outcom > .z;
    IF &__expo    > .z;
    /* repeat the exclusions for each exposure*/
    /* variable in the base model             */
    %DO i = 1 %TO &__varkt;
        IF &&__bv&i   > .z;
    %END;
    CALL SYMPUT
("__vfmt",TRIM(PUT(VFORMATN(&__expo),$CHAR8.)));
  RUN;

/* create a PROC FREQ output data set that    */
/* will be used to attain the total number of */
/* levels in the outcome variable             */
 PROC FREQ DATA=__logis;
   TABLES &__outcom / NOPRINT OUT=__oc;
 RUN;

/* Create two macro variables that contain    */
/* the levels of the outcome variable. Set    */
/* marco variables for the n's of the two     */
/* levels of the outcome variable put the     */
/* total number of outcome levels into a macro*/
```

```
/* variable                                    */
DATA _NULL_;
  SET __oc NOBS=obs_tot;
  IF _N_=1 THEN CALL SYMPUT
     ('__oc_kt',LEFT(PUT(obs_tot ,4.)));
  %DO i= 1 %TO 2;
      IF &i=_N_ THEN DO;
         %GLOBAL __oc&i.;
CALL SYMPUT("__oc&i.",LEFT(PUT(&__outcom,4.)));
CALL SYMPUT("__oc_kt&i.",LEFT(PUT(COUNT,5.)));
      END;
  %END;
RUN;

/* create a PROC FREQ output data set that    */
/* will be used to attain the total number of */
/* levels in the exposure variable            */
PROC FREQ DATA=__logis;
  TABLES &__expo / NOPRINT OUT=__levels;
RUN;

/* put the total number of levels of exposure */
/* variable  into a macro variable            */
DATA _NULL_;
  IF 0 THEN SET __levels NOBS=obs_tot;
  IF _N_=1 THEN DO;
     CALL SYMPUT
     ('__exp_kt' , LEFT(PUT(obs_tot ,4.)));
     STOP;
  END;
RUN;

/* Create two sets of macro variables. The    */
/* first set contains the values of the levels*/
/* in the exposure variable. The second set   */
/* contains the values of the formats for each*/
/* level in the exposure variable             */
%DO i= 1 %TO &__exp_kt;
    %GLOBAL __exp&i. __fmt&i.;
    DATA _NULL_;
      SET __levels NOBS=obs_tot;
      IF &i=_N_ THEN DO;
         CALL SYMPUT
         ("__exp&i.",LEFT(PUT(&__expo,4.)));
         CALL SYMPUT
    ("__fmt&i" ,LEFT(PUT(&__expo,&__vfmt..)));;
         IF &__expo=&__ref THEN
            CALL SYMPUT
    ("__reffmt" ,LEFT(PUT(&__expo,&__vfmt..)));;
      END;
    RUN;
  %END;

/* determine the crude p-value for the trend  */
/* association                                */
  QUIT;
  ODS OUTPUT  ChiSq=ods_chi;

  PROC FREQ DATA=__logis;
     TABLES &__expo*&__outcom / CHISQ MISSPRINT;
  RUN;
  ODS OUTPUT CLOSE;
  DATA _NULL_;
    SET  ods_chi;
    IF statistic='Mantel-Haenszel Chi-Square'
    THEN CALL SYMPUT
       ('trend_p',LEFT(PUT(PROB,5.4)));
  RUN;

/* determine the crude association for each   */
/* comparison level                           */
/* of the exposure variable with the referent */
/* level when crossed with outcome variable in*/
/* 2x2 frequency tables                       */
%IF &__ctn NE 1 %THEN %DO;
   %DO i= 1 %TO &__exp_kt;
      QUIT;

   /* output datasets which contain the       */
   /* frequencies and percents for each cell in*/
   /* the 2x2 tables as well as the crude     */
   /* associations for each 2x2 table         */
    ODS OUTPUT  ChiSq=ods_chi
                CROSSTABFREQS=ods_freq;
    PROC FREQ DATA=__logis;
      WHERE &__expo=&&__exp&i. | &__expo=&__ref;
      TABLES &__expo*&__outcom / CHISQ MISSPRINT;
    RUN;
    ODS OUTPUT CLOSE;

    /* generate a macro variable with the crude*/
    /* association p-value.                    */
    DATA _NULL_;
      SET  ods_chi;
      IF statistic='Mantel-Haenszel Chi-Square'
      THEN CALL SYMPUT
           ('crude_p',LEFT(PUT(PROB,5.4)));
    RUN;

    /* generate macro variables which contain  */
    /* the frequencies and percents for each   */
    /* appropriate cell in the 2x2  tables     */
    DATA _NULL_;
      SET ods_freq;
      IF &__outcom=&__oc1 AND &__expo=&&__exp&i.
      THEN CALL SYMPUT
           ('__cell3',LEFT(PUT(FREQUENCY,5.0)));
      ELSE
      IF &__outcom=&__oc2 AND &__expo=&&__exp&i.
      THEN CALL SYMPUT
           ('__cell4',LEFT(PUT(FREQUENCY,5.0)));
    RUN;

    /* Create a one observation data set which  */
    /* contains all frequencies, percents and   */
    /* p-values desired. Also add variables     */
    /* for formats, values and variable names   */
    /* for the outcome and exposure variables   */
    DATA __out&i;
      LENGTH exposure       $ 32;
      LENGTH exposure_label $ 42;
```

```
    LENGTH outcome_label  $ 42;
    exposure="&__expo";
    n_controls=&__cell3;
    n_cases=&__cell4;
    IF &&__exp&i=&__ref THEN crude_p_value=.;
       ELSE crude_p_value=&crude_p;
    pct_controls = (n_controls/&__oc_kt1)*100;
    pct_cases = (n_cases/&__oc_kt2)*100;
    exposure_label="&&__fmt&i.";
    exposure_value=&&__exp&i.;
    outcome="&__outcom";
    outcome_label="&__oclab";
    FORMAT crude_p_value pvalue.;
    LABEL exposure       =
"Name of Independent Exposure Variable"
          exposure_label =
"Exposure Label(Class/Dichotomous Level Format)"
          exposure_value =
"Dichotomous/Class: Value of Exposure Level"
          outcome        =
"Name of Dependent Outcome Variable"
          outcome_label  =
"Label of Outcome Variable";
  RUN;
%END;

/* combine all the data sets containing the   */
/* statisitics for each of the 2x2 comaprisons*/
  DATA __fstats;
    SET %datasets;
    LABEL crude_p_value =
  'PROC FREQ CMH p-value'
          n_cases       =
  'PROC FREQ Frequency of Cases(Column 2)'
          n_controls    =
  'PROC FREQ Frequency of Controls(Column 1)'
          pct_cases     =
  'PROC FREQ Percent of Cases'
          pct_controls  =
  'PROC FREQ Percent of Controls'
          ;
  RUN;
%END;

/* If the exposure variable passed in the     */
/* macro call contains more than 2 levels,    */
/* then run the model with the trend variable.*/
%IF &__exp_kt>2 %THEN %DO;
  QUIT;
  ODS OUTPUT parameterestimates=__param
             oddsratios=__odds;
  RUN;
  /* include the base model expsoure variables*/
  /* as well as the additional exposure       */
  /* variable for which statisitics are being */
  /* genreated in the macro                   */
  PROC LOGISTIC DATA=__logis DESCENDING;
    MODEL &__outcom = &__basevs &__expo;
      OUTPUT OUT=logout;
    RUN;
  ODS OUTPUT CLOSE;

  DATA __param;
    SET __param;
        IF VARIABLE="&__expo";
    LABEL VARIABLE =
"Name of Exposure Variable";
    RENAME VARIABLE = exposure;
  RUN;

  DATA __odds;
    SET __odds;
    IF UPCASE(EFFECT)=UPCASE("&__expo");
    LABEL EFFECT =
"Name of Exposure Variable";
      RENAME EFFECT=exposure;
  RUN;

  DATA __trend;
    LENGTH exposure $ 32;
    LENGTH exposure_type $ 11;
    LENGTH exposure_label $ 42;
    LENGTH outcome_label $42;
    MERGE __param __odds;
      BY exposure;
    Exposure_label="&__explab";
    n_controls=&__oc_kt1;
    n_cases=&__oc_kt2;
    pct_controls=100;
    pct_cases=100;
    %IF &__ctn = 1 %THEN
        exposure_type="Continuous";
    %ELSE exposure_type="Trend";;
    crude_p_value=&trend_p;
    outcome="&__outcom";
    outcome_label="&__oclab";
    FORMAT crude_p_value pvalue.;
    LABEL oddsratioest   =
'PROC LOGISTIC Adjusted Odds Ratio Estimate'
          probchisq      =
'PEOC LOGISTIC Adjusted p-value'
          exposure_type  =
'Dichotomous/Trend/Class/Continuous'
          crude_p_value  =
'PROC FREQ CMH p-value'
          n_cases        =
'PROC FREQ Frequency of Cases(Column 2)'
          n_controls     =
'PROC FREQ Frequency of Controls(Column 1)'
          pct_cases      =
'PROC FREQ Percent of Cases'
          pct_controls   =
'PROC FREQ Percent of Controls'
          exposure_label =
"Exposure Label(Class/Dichotomous Level Format)"
          outcome        =
"Name of Dependent Outcome Variable"
          outcome_label  =
"Label of Outcome Variable";
    DROP DF ESTIMATE STDERR WALDCHISQ;
    RUN;
%END;
```

```
/* Run the PROC LOGISTIC with the CLASS option*/
/* for both the dichotomous variables and     */
/* variables containing more than 2 levels.   */
/* Statistics will be genreated and kept for  */
/* each level determined to be added by the   */
/* CLASS statement.                           */
   QUIT;
%IF &__ctn NE 1 %THEN %DO;

  ODS OUTPUT parameterestimates=__param
             oddsratios=__odds;
  RUN;
  /* request class variables for the modeling */
  /* of the additional exposure variable in   */
  /* the model.                               */
  PROC LOGISTIC DATA=__logis DESCENDING;
    CLASS &__expo (PARAM=REF REF="&__reffmt"
                   ORDER=INTERNAL);
    MODEL &__outcom = &__basevs &__expo;
    OUTPUT OUT=logout;
    RUN;
  ODS OUTPUT CLOSE;

  RUN;

  DATA __param;
    SET __param;
    IF UPCASE(variable)=UPCASE("&__expo");
    RENAME CLASSVAL0 = exposure_label;
    %DO i = 1 %TO &__exp_kt;
        IF CLASSVAL0="&&__fmt&i" THEN
           exposure_value=&&__exp&i.;
    %END;
    LABEL CLASSVAL0      =
"Exposure Label(Class/Dichotomous Level Format)"
          exposure_value =
"Dichotomous/Class: Value of Exposure Level";
    DROP VARIABLE DF ESTIMATE STDERR WALDCHISQ;
  RUN;

  DATA __odds;
    SET __odds;
    found_var=INDEX
    (UPCASE(effect),UPCASE(TRIM("&__expo")));
    IF found_var>=1;

    %DO i = 1 %TO &__exp_kt;
  /* continue if the class value is not the   */
  /* referent level                           */
        IF &&__exp&i ^= &__ref THEN DO;
           found_label=
           INDEX(EFFECT,TRIM("&&__fmt&i"));
           IF found_label>=1 THEN
              exposure_value=&&__exp&i.;
        END;
    %END;
    LABEL exposure_value =
"Dichotomous/Class: Value of Exposure Level";
    DROP EFFECT found_var found_label;
  RUN;

  PROC SORT DATA=__param;
    BY exposure_value;
  RUN;
  PROC SORT DATA=__fstats;
    BY exposure_value;
  RUN;
  PROC SORT DATA=__odds;
    BY exposure_value;
  RUN;

  DATA __cstats;
    LENGTH exposure_type $ 11;
    MERGE __fstats __param __odds;
      BY exposure_value;
    IF &__exp_kt>2 THEN exposure_type='Class';
    ELSE IF &__exp_kt=2 THEN
         Exposure_type='Dichotomous';
    LABEL oddsratioest =
'PROC LOGISTIC Adjusted Odds Ratio Estimate'
          probchisq     = '
PROC LOGISTIC Adjusted p-value'
          exposure_type      =
'Dichotomous/Trend/Class/Continuous';
  RUN;
%END;

/* If there are more than 2 levels to the     */
/* exposure variable passed in the macro call,*/
/* then the trend model was performed.  Add   */
/* the statistics from the trend model to the */
/* data set containing the statistics from the*/
/* model with the 'CLASS' statement.          */
%IF &__exp_kt>2 AND &__ctn = 0 %THEN %DO;
   DATA __cstats;
     SET   __trend __cstats;
   RUN;
%END;
%ELSE %IF &__ctn = 1 %THEN %DO;
     DATA __cstats;
     SET   __trend ;
   RUN;
%END;

%IF %SYSFUNC(EXIST(__sumary)) %THEN %DO;
    DATA __sumary;
      SET __sumary __cstats;
    RUN;
%END; %ELSE %DO;
    DATA __sumary;
      SET __cstats;
    RUN;
%END;
RUN;
%MEND summary;
```

Appendix 2. Example SAS program

```
PROC FORMAT LIBRARY=library;
RUN;
```

```
%LET log_vars=%STR(age_16_to_20 age_21_to_30
              age_31_to_40 age_51_to_60
              age_61_to_70 age_71_to_80
              smoker_past smoker_current
              asthma atopy);
RUN;

/* Include the file that contains the code for*/
/* the macro.                                 */
%INCLUDE 'd:\studies\macros\summary.sas';

/* The dataset '__sumary' will need to be     */
/* removed from the working library if it     */
/* has been created by precious runs of the   */
/* MACRO summary.                             */
PROC DATASETS;
  DELETE __sumary;
  RUN;
QUIT;

/*  The following parameters are required for */
/* the macro to execute(If the exposure       */
/* parameter is preceded by a '#' then the    */
/* variable will be treated as continuous and */
/* will not generate CLASS analysis)          */
/*                                            */
/*dataset outcome  exposure referent BaseModel*/
/*        variable variable level    variables*/
/* -------------------------------------------*/
%summary
(sasdset, WHEEZE,  STATE,    0,      &log_vars)
%summary
(sasdset, WHEEZE,  FARMSIZE, 2,      &log_vars)
%summary
(sasdset, WHEEZE,  #WEIGHT,  2,      &log_vars)

/* Print the summary table using Data Null    */
%MACRO printit(titlea, titleb, filenm);
DATA _NULL_;
  LENGTH _previous_ $ 32;
  SET __sumary;
  FILE "d:\studies\sd2\&filenm" LS=140;
  IF _N_=1 THEN DO;
     _previous_=exposure;
     PUT "PROGRAM SESUG.SAS.sas:  &sysdate";
     PUT 'Models adjusted for Age, Past Smoking'
         ', Current Smoking, and Asthma-Atopy '
         'combined variable';
    PUT;
    PUT;
    PUT "&titlea." @51 "Test Type" @66
        " WHEEZE No    WHEEZE Yes      "
        "P-Value          OR      95% CL";
    PUT "&titleb." @66
        "    N    %        N    %      "
        "          Crude  Adjusted";
    PUT '-------------------------------------'
        '-------------------------------------'
        '-------------------------------------'
        '----------------' ;
  END;
  IF exposure^=_previous_ THEN PUT;
  PUT exposure $ 1-9 exposure_label $ 10-49
      exposure_type $ 51-62 n_controls 66-70 @72
      pct_controls 4.1 n_cases 80-84 @86
      pct_cases 4.1 @96 crude_p_value @104
      probchisq @115 oddsratioest @122 lowercl
      @129 uppercl;
  _previous_=exposure;
  RETAIN _previous_;
  RUN;
%MEND printit;

/*******************************************/
/* Print the summary table to an output text  */
/* file.                                      */
/*       Exposure  Exposure        Output     */
/*       Title 1   Title 2         Text File  */
/*       -------- -------------- -------------*/
%printit(Variable ,Name     Label, %STR(a.txt));
```

References

SAS Institute Inc. (1990). SAS/STAT User's Guide, Version 6, Fourth Edition, Cary, NC: SAS Institute Inc.

SAS Institute Inc. (1990). SAS/MACRO LANGUAGE, Version 6, Fourth Edition, Cary, NC: SAS Institute Inc.

Acknowledgments

The authors would like to thank David Shore for their assistance and review of the methods, coding, and manuscript.

Contact Information

Stuart Long, Rebecca Darden
Westat, Inc.
1009 Slater Road, Suite 120
Durham, NC 27703